ABSTRACT
The SAS macro language gives us the power to create tools that to a large extent can think for themselves. How often have you used a macro that required your input and you thought to yourself “Why do I need to provide this information when SAS already knows it?” SAS may well already know what you are being asked to provide, but how do we direct our macro programs to self-discern the information that they need? Fortunately there are a number of functions and other tools within SAS that can intelligently provide our programs with the ability to find and utilize the information that they require.

If you provide a variable name, SAS should know its type and length; given a data set name, the list of variables should be known; given a library or libref, the full list of data sets that it contains should be known. In each of these situations there are functions that can be utilized by the macro language to determine and return these types of information.

Given a libref these functions can determine the library’s physical location and the list of all the data sets it contains. Given a data set they can return the names and attributes of any of the variables that it contains. These functions can read and write data, create directories, build lists of files in a folder, and even build lists of folders.

Maximize your macro’s intelligence; learn and use these functions.

KEYWORDS
Metadata, macro language, DATA step functions, %SYSFUNC, OPEN, ATTRN, ATTRC, FETCH, CLOSE

INTRODUCTION
There are a number of ways for a macro to determine information about the operating system, the SAS environment, libraries and the files that they contain, data sets and their variables, and variable attributes and the values that they contain (Carpenter, 2016). Among others these include the use of the X statement (and related statements and functions), DICTIONARY tables, SASHELP views, data sets created by PROC CONTENTS, system options, and automatic macro variables (Carpenter and Rosenbloom, 2016). Of these various approaches DATA step functions tend to be fastest and as a bonus they support the creation of user defined macro functions.

There are also a number of DATA step functions that can be used to obtain information about the SAS environment. Although used less often in the DATA step itself, these functions are invaluable to the macro programmer. Collectively they are often referred to as metadata functions, because they tend to return information about data sets and locations. In fact they actually have a much broader list of capabilities that extends beyond the metadata. These functions can also read and write data; create, count, and eliminate both directories and files; return path and location information; and much more. Of these functions, some of the more commonly used are highlighted here.
Within the macro language, these DATA step functions are typically accessed using the %SYSFUNC or %QSYSFUNC macro functions. For each of these functions, the first argument is the DATA step function (along with its own arguments) that is to be executed, and an optional format in the second argument to control the appearance of the value returned by the function in the first argument. In this example of a TITLE statement, the DATE function is called. It returns the current date as a SAS date, and this value is then formatted using the WORDDATE18. format. The generated TITLE statement will then contain the formatted date.

Most of the macros in this paper which demonstrate these DATA step functions are themselves macro functions. As macro functions the macro itself returns a value. If you are unfamiliar with the coding techniques used to create user written macro functions, you may want to review Carpenter, 2002 for more detail than will be described here.

**VIEWING DATA SET METADATA**

Metadata is data about the data set. Each SAS data set automatically stores, as a part of the data set, information about the data set itself, hence metadata. This is the information that you see when you execute a PROC CONTENTS. Although of interest to the DATA step programmer, access to the metadata can provide huge benefits to the macro programmer.

Typically the CONTENTS procedure is used just to display the metadata, however it can also be used to store the metadata in a data set as well. This data set has one row per variable and it contains information on the individual variables as well as data set specific information such as the number of observations in the data set. A portion of the metadata data set generated by CONTENTS for the SASHELP.CLASS data set is shown below.

```sas
proc contents data=sashelp.class out=classvar noprint;
run;
```

One row per variable per data set can be obtained by using `data=sashelp._all_`. This is similar to the information contained in the view SASHELP.VCOLUMNNS which is shown in the next example.
Without executing a PROC CONTENTS, similar information can be found in the SASHELP view VCOLUMN, however this view has one row per variable per data set per library known to SAS.

Almost the same information can also be surfaced through the use of the SQL DICTIONARY table COLUMNS. The DESCRIBE statement can be used to list the names of the columns, while the SELECT statement can write the values held in the table.

Both SASHELP.VCOLUMN and DICTIONARY.COLUMNS are created when they are requested. This means that they will always contain current information. This also means that there can be a significant wait while these tables are created. The metadata functions described in this paper do not have this limitation.

Viewing the metadata is oftentimes sufficient, but what if during the execution of a program we need to gather some metadata item and use it in the program? How can we have our program do this dynamically? Yes we can! Enter the DATA step's metadata functions.

**DATA SET METADATA FUNCTIONS**

The DATA step metadata functions operate directly against the metadata of a SAS data set. This means that we can avoid the expenditure of resources that are needed to create the SASHELP.VCOLUMN view or the DICTIONARY.COLUMNS table. It also means that through the use of these functions we can dynamically gather and use metadata in our programs directly.

Since these are DATA step functions, when they are to be used with the macro language you will need to invoke them through the use of the %SYSFUNC or %QSYSFUNC macro functions. This allows the DATA step function to execute during macro execution and to return the value to the macro facility.
**Opening the Metadata – making it available for our use**

The data set’s metadata is not automatically available for our use. Before we can access it we need to get “permission” to look at the data set’s descriptor record which contains the metadata. We gain this access through the use of the OPEN function. This function checks to see if the data set is currently available for our use. If the data set is available, the OPEN function returns a non-zero data set identifier. We never really care what the identifier is, as long as it is not zero (which would indicate that we have been denied access to the data set). After we have utilized the information in the metadata we need to release the data set so that other programs or programmers can use it. We do this by closing our access through the use of the CLOSE function.

The macro `%META` shown here is just a shell that highlights the use of the OPEN and CLOSE functions. These functions will almost always be present when working with a data set’s metadata.  

1. The OPEN function is used to gain access to the data set’s metadata. When the data set is successfully opened for use, it returns a data set identifier, which, in this macro, is stored in the local macro variable &DSID. When the data set is not opened successfully the value returned is a 0. This allows us to test whether or not the metadata is available.  

2. After we have finished using the data set, it is closed with the CLOSE function. This function returns a 0 for success. In this example macro the identifier is also cleared by replacing it with the value returned by the CLOSE function.  

The data set identifier that is returned by the OPEN function is used by many of the other metadata functions. Having a unique identifier associated with a given data set is necessary as you may wish to open the metadata of more than one data set at a time. Usually this identifier will be stored in a local macro variable. Remember although we need to store the value of the identifier, the value itself is rarely of specific interest.

**Variable Information Functions**

As the name implies, variable information functions return information about the variables in the data set. These functions are commonly used when you need to ask questions such as; “Is the ABC variable on this data set?”, “Does the ABC variable have a format?”, and “What is the type of the ABC variable?”. The functions of this type all start with the letters ‘VAR’, however not all functions that start with ‘VAR’ fit the category.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Returns the:</th>
</tr>
</thead>
<tbody>
<tr>
<td>VARFMT</td>
<td>Variable’s assigned format.</td>
</tr>
<tr>
<td>VARINFMT</td>
<td>Variable’s informat.</td>
</tr>
<tr>
<td>VARLABEL</td>
<td>Label of a variable.</td>
</tr>
<tr>
<td>VARLEN</td>
<td>Length of a variable.</td>
</tr>
<tr>
<td>VARNAME</td>
<td>Name of a SAS data set variable.</td>
</tr>
<tr>
<td>VARNUM</td>
<td>Number of a variable’s position.</td>
</tr>
<tr>
<td>VARTYPE</td>
<td>Data type of a SAS data set variable (C or N)</td>
</tr>
</tbody>
</table>

Table 1: Variable Information Functions

One of the most commonly accessed metadata attributes is information about variable names and the existence of variables on a data set. The two functions that apply specifically to the variable name is VARNAME and VARNUM. Given a variable number (position on the PDV), the VARNAME function returns the name of the variable. The VARNUM function is the opposite as it provides the position number given the variable name.
The macro %VAREXIST checks to see if a specific variable exists on a specified data set. The name of the variable of interest is passed to the VARNUM function, which in turn returns the variable’s position. If the variable does not exist on the data set, the VARNUM function returns a 0.

The first argument of the VARNUM function is the data set identifier (obtained from the OPEN function), and the second is the name of the variable of interest. In this example, the value returned by the VARNUM function is stored in the local macro variable &VNUM.

The macro %VAREXIST is written as a macro function. The value returned by VARNUM is passed out of the macro, and the macro is said to resolve to the returned value. This means that the macro call can be used in other statements, such as an %IF, to make decisions about further processing.

Data Set Attribute Functions

When a data set is created or modified, a number of attributes of the data set are stored in the metadata. These attributes include things like:

- when the data set was created
- how many variables it contains
- how many observations it contains (there can be more than one answer)
- data set size
- status of indexes, WHERE clauses, and passwords.

The two functions that return this type of attribute information are the ATTRC (returns character information) and ATTRN (returns numeric information) functions. Each of these functions can return a number of different attributes that can be selected by the user. These requests are made by specifying an attribute as the second argument to the function. Some of the available attribute request options are shown in Table 2.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Attribute Request</th>
<th>Returns the:</th>
</tr>
</thead>
<tbody>
<tr>
<td>ATTRC</td>
<td>COMPRESS</td>
<td>Compression status</td>
</tr>
<tr>
<td></td>
<td>ENGINE</td>
<td>Name of the engine used to create the data set</td>
</tr>
<tr>
<td></td>
<td>LABEL</td>
<td>Data set’s label</td>
</tr>
<tr>
<td></td>
<td>LIB</td>
<td>Location of the data set, the Library (libref) name</td>
</tr>
<tr>
<td></td>
<td>MEM, DNAME</td>
<td>Name of the data set</td>
</tr>
<tr>
<td></td>
<td>SORTEDBY</td>
<td>List of BY variables used to sort the data</td>
</tr>
<tr>
<td></td>
<td>TYPE</td>
<td>Data set type</td>
</tr>
<tr>
<td>ATTRN</td>
<td>CRDTE, MODTE</td>
<td>Datetime the data set was created or last modified</td>
</tr>
<tr>
<td></td>
<td>ISINDEX, INDEX</td>
<td>Status on indexes for this data set</td>
</tr>
<tr>
<td></td>
<td>NDEL, NLOBS, NLOBSF, NOBS</td>
<td>Number of observations in the data set based on various conditions (primarily whether or not to count those marked for deletion)</td>
</tr>
<tr>
<td></td>
<td>NVAR</td>
<td>Number of variables in the data set</td>
</tr>
<tr>
<td></td>
<td>various</td>
<td>Read, write, and alter password status</td>
</tr>
</tbody>
</table>

Table 2: Data Set Attribute Functions
The use of the ATTRN function is demonstrated in the macro %VARLIST which creates a list of variable names of a specified type (numeric or character) for the data set of interest. The user passes in the data set name and whether or not to select a specific type of variable. The macro then returns the list of variables (or a blank if no variables meet the criteria).

The ATTRN function is used with the NVARS attribute to request the number of variables in the data set. The index (&I) for the %DO loop will then cycle from 1 to the number of variables. The loop index (&I) is used to identify a specific variable in both the VARTYPE function and the VARNAME function.

The VARTYPE function will return the type of the &Ith variable. The returned value is either a C or N and this value is compared to the requested type stored in &VTYPE.

The VARNAME function returns the name of the &ith variable. The name of each variable that meets the criteria is added to the growing list stored in &VARLIST.

Once the list of variables has been created it is passed out of the macro.

The SAS Log to the right shows how the %VARLIST macro could be used to create a list of variable names of varying types.
Because the number of observations in a data set is stored in the data set’s metadata, the fastest way to determine the number of observations is to access the metadata directly using the macro language. The data set is opened and the ATTRN function is used with the NLOBS (number of non-deleted observations) to return the number of observations contained in the data set.

The data set is opened. If the data set cannot be opened, this macro returns a dot (.) for the number of observations.

The NLOBS attribute is used with the ATTRN function to return the number of non-deleted observations. This takes into account any observations that may have been marked for deletion during an interactive session using PROC FSEDIT or a similar tool.

The data set is closed after retrieving the value of interest.

If the data set is not available, a warning along with the reason (using the SYSMSG function) is written to the SAS Log.

The number of observations is returned by the %OBSCNT macro.

Notice that all the macro variables created by the %OBSCNT macro are forced onto the local symbol table.

READING DATA USING FUNCTIONS

Although not a common requirement, it is possible to both read and write data held in a SAS data set using the macro language. Once a data set has been opened, you can read observations both sequentially and using random access techniques.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Action:</th>
</tr>
</thead>
<tbody>
<tr>
<td>CUROBS</td>
<td>Current observation number</td>
</tr>
<tr>
<td>FETCH</td>
<td>Reads the next non-deleted observation from a SAS data set into the Data Set Data Vector (DDV)</td>
</tr>
<tr>
<td>FETCHOBS</td>
<td>Reads a specified observation from a SAS data set into the Data Set Data Vector (DDV)</td>
</tr>
<tr>
<td>GETVARN</td>
<td>Returns the value of a numeric variable</td>
</tr>
<tr>
<td>GETVARC</td>
<td>Returns the value of a character variable</td>
</tr>
<tr>
<td>NOTE/DROPNOTE</td>
<td>NOTE stores a unique observation ID number</td>
</tr>
<tr>
<td>POINT</td>
<td>Locates the observation identified by NOTE</td>
</tr>
<tr>
<td>REWIND</td>
<td>Returns the observation pointer to the beginning of the data set</td>
</tr>
<tr>
<td>CALL SET</td>
<td>Links data set variables to macro variables of the same name</td>
</tr>
</tbody>
</table>

Table 3: Data Access Functions
The macro `%SYMCHECK` (Carpenter, 2016, Exp. 9.2.1b) can be used to determine if a given macro variable is currently defined on a specific symbol table. The view SASHELP.VMACRO is used as input for the macro. As it is opened, a WHERE clause is used to limit the read to the specific row of interest.

```sas
%macro symcheck(mscope,mvname);
  %* Determine if a specific macro variable
  %* has been defined in a specific scope.;
  %local fetchrc dsnid rc;
  %let rc = 0;
  %let dsnid = %sysfunc(open(sashelp.vmacro (where=(scope=%upcase("&mscope") and name=%upcase("&mvname"))),i));
  %let fetchrc = %sysfunc(fetch(&dsnid,noset));
  %if &fetchrc eq 0 %then %let rc=1;
  %let dsnid = %sysfunc(close(&dsnid));
  &rc
%mend symcheck;
```

1. The view is opened using a WHERE= data set option to limit the number of rows available – there should be exactly one row if the macro variable of interest exists within the specified scope.
2. The FETCH function is used to determine if a row meets the criteria in the WHERE clause. The NOSET attribute specifies that the values stored in the view are not to be transferred to macro variables.
3. A successful read indicates that the row exists, and the FETCH function returns a 0 for success.
4. &RC will contain a 0 if the specified macro variable does not exist, and a 1 if it does. Regardless this value is returned by the macro.

The `%M_ALL_DATA` macro shown next, can be used to mimic the functionality of the DATA step’s CALL EXECUTE routine. In this macro both the FETCHOBS function and the SET routine are used (Carpenter, 2016, Exp. 9.2.2d) to build the macro variables. In fact it creates a macro variable for each variable in the named data set, and these macro variables are then populated using the data in the data set.

```sas
%macro M_all_data(dsn=);
  %local dsid i nobs;
  %let dsid = %sysfunc(open(&dsn));
  %let nobs= %sysfunc(attrn(&dsid,nlobs));
  %syscall set(&dsid);
  %do i = 1 %to &nobs;
    %let rc=%sysfunc(fetchobs(&dsid,&i));
    %* Local process goes here; 
    %put ***** Observation &i *********;
    %put _local_;
  %end;
  %let dsid = %sysfunc(close(&dsid));
%mend M_all_data;
```

5. The SET routine is used to match the names of the variables in the data set with the names of the macro variables to be created.
6. The %DO loop is used to step through all of the observations in the incoming data set (there will be NLOBS of them).
7. A FETCHOBS function is used to read the next observation from the data set. Because of the use of the SET routine 5. the values of the variables read from the data set are automatically written to corresponding macro variables of the same name.
8. A typically usage of this macro would be to call another macro at this point that would utilize the newly created observation specific macro variables.
For the first observation in the SASHELP.CLASS data set, the SAS Log shows that a local macro variable has been created for each variable in the incoming data set. The values of those macro variables correspond to the values of the data set variables.

The important thing to note in this example is that for each variable in the data set, a macro variable has been created with the same name and value as the variable. This can be very advantageous when there are a large number of macro variables that need to be created, and even more especially so if we do not necessarily know the names of the macro variables. This is demonstrated in the following example which calls the macro %SHOE_RPT once for each observation in a control file, and demonstrates the utility of this approach.

A control file is constructed with the desired attributes. In this case the two variables in the WORK.SHOE_RPT data set will become the parameters that will be used with a reporting macro named %SHOE_RPT. This macro depends on two macro variables (&REGION and &PRODUCT).

To take advantage of the control data set (WORK.SHOE_RPT) the macro %M_ALL_DATA has been slightly modified at ➒. It will now call a macro with the same name as the name of the control data set (SHOE_RPT), and since it is called inside the %DO loop, it will be called once for each observation. For slippers sold in Asia the report to the left is generated.

In a blog Leonid Batkhan (Batkhan, 2016) uses the SET routine along with the FETCH function to read variable attributes.
WORKING WITH SAS FILES (LIBRARIES, DATA SETS, AND CATALOGS)

Creating and maintaining libraries of SAS files, like data sets and catalogs, can also be managed using functions. While these functions do not use the metadata of SAS data sets, they return information about the libraries—effectively metadata about the directory. These functions are often used in conjunction with those already described.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Action:</th>
</tr>
</thead>
<tbody>
<tr>
<td>CEXIST, EXIST</td>
<td>Whether or not the entity exists</td>
</tr>
<tr>
<td>LIBNAME</td>
<td>Establish a libref</td>
</tr>
<tr>
<td>LIBREF</td>
<td>Check for libref existence</td>
</tr>
<tr>
<td>PATHNAME</td>
<td>Return the physical path for a libref or fileref</td>
</tr>
<tr>
<td>RENAME</td>
<td>Rename a data set or catalog</td>
</tr>
</tbody>
</table>

Table 4: SAS File Functions

When you need to work with libraries of SAS files (data sets or catalogs) or with SAS files as entities, the functions in Table 4 can be of assistance. The LIBNAME and LIBREF functions can be used to establish or check the existence of a library. While the CEXIST and EXIST functions are used to determine whether or not a catalog or data set exists.

The PATHNAME function returns a physical path given a libref or a fileref. In this example a series of data sets are to be copied from a SAS data directory with a libref of PROJMETA into an Excel workbook using the PCFILES engine. Because we want to create the workbook in the same directory as the original data, the PATHNAME function is used to return the current path to the libref PROJMETA.

```sas
libname toxls pcfiles ➊  path="%qsysfunc(pathname(projmeta))\MyExcelData.xls";
proc datasets nolist;
   copy inlib=sashelp outlib=toxls; ⃣
      select class heart prdsale; ❼
   quit;
libname toxls clear; ➋
```

➊ The PCFILES interpretation engine is selected for the new libref.
⃣ The PATHNAME function is used to return the physical path used by the PROJMETA libref.
❼ The PROC DATASETS COPY statement is used to point to the incoming and outgoing librefs.

The SELECT statement specifies which data sets are to be copied.
➋ The libref must be cleared before the new workbook can be used.
In the previous example, because an existing libref is used, we know that its associated folder also exists. Often we will be given a folder or path to use and we will need to either verify that it is valid or to create if it does not already exist. Given a path/location, the %CHECKLIB macro uses the LIBNAME and LIBREF functions to establish or clear a libref.

```sas
%macro checklib(libref=,libpath=);
  %local rc;
  %if &libpath= %then %do;
    %* Clear this libref;
    %let rc=%sysfunc(libname(&libref)); ➏
  %end;
  %else %if %sysfunc(libref(&libref)) ne 0 %then %do;
    %* Establish this libref;
    %let rc=%sysfunc(libname(&libref,&libpath)); ➑
    %put %sysfunc(sysmsg());
  %end;
  %else %do;
    %sysfunc(sysmsg()); ➉
    %put WARNING: LIBREF not reassigned;
  %end;
%mend checklib;
```

When only a libref is passed to the LIBNAME function, the libref is cleared. ➏ The LIBREF function determines whether or not the libref is defined. The function returns a non-zero value if the libref does not exist. ➐ When a second argument is present and the libref does not already exist, the LIBNAME function is used to establish the libref. ➑ The SYSMSG function is used to return error messages when the macro is unable to establish the libref.

WORKING WITH DIRECTORIES

There are a number of functions available that have been designed to work with directories or folders in much the same way that the functions discussed earlier work with data sets. These functions can be used to create new folders as well as to read the names of the files within a folder.

Common approaches to working with folders include the use of the X statement or one of its equivalents (%SYSTASK, CALL SYSTEM, etc.). Because these techniques tend to be OS dependent, they require the programmer to understand the language/commands of the OS sub session. For the Windows OS the sub session, these are DOS commands. You can learn more about DOS commands by using the help command or by pairing help with the name of a command to get command specific help.

Here the X statement is used to create a Windows directory (using the MD command), and to write a list of SAS programs and data sets to a text file (using the DIR command).

```sas
x md "c:\temp\test"; ➊
x dir "c:\temp\*.sas" /o:n /b > "c:\temp\test\pgmlist.txt"; ➋
```

➊ The MD command is used to create a directory. ➋ The DIR command is used to write the names of the files that contain SAS in the extension to the text file PGMLIST.TXT.
The DATA step’s file and directory functions allow us to accomplish the same basic types of tasks as the X statement without resorting to OS specific syntax and without stepping out of the macro language. Some of the more common functions of this type are shown in Table 5.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Action</th>
</tr>
</thead>
<tbody>
<tr>
<td>DOPEN and DCLOSE</td>
<td>Open and close a directory</td>
</tr>
<tr>
<td>DCREATE</td>
<td>Create a directory or sub folder</td>
</tr>
<tr>
<td>DREAD</td>
<td>Read the names of items within a directory</td>
</tr>
<tr>
<td>DNUM</td>
<td>Returns the number of items in a directory</td>
</tr>
<tr>
<td>FILEEXIST</td>
<td>Checks existence of a file or directory</td>
</tr>
<tr>
<td>MOPEN</td>
<td>Open a file within a directory</td>
</tr>
</tbody>
</table>

Table 5: Directory and File Functions

The %CHECKLOC macro utilizes the FILEEXIST function to determine if a directory exists, and if it does not exist, the DCREATE function is used to create the directory. This macro is written as a macro function that returns the full path of the directory (whether it already exists or if it is created).

```macro
%macro CheckLoc(DirLoc=, DirName=); ③
   %* if the directory does not exist, make it;
   %if %sysfunc(fileexist("&dirloc\&dirname"))=0 %then %do; ④
      %put Create the directory: ";&dirloc\&dirname";
      %* Create the directory;
      %sysfunc(dcreate(&dirname,&dirloc)) ⑤
   %end;
   %else %do;
      %put The directory "&dirloc\&dirname" already exists;
      &dirloc\&dirname ⑥
   %end;
%mend checkloc;
```

③ The upper portion of the path along with the folder name is passed into the macro. ④ The FILEEXIST function is used to determine whether or not the specified folder already exists. This function returns a 0 when the folder does not already exist. ⑤ The DCREATE function can be used to make a directory. The first argument is the directory name and the second argument is the upper portion of the path. Notice that the call to %SYSFUNC stands alone and is not a part of a complete statement. Because the DCREATE function returns the full path, this line will resolve to the full path, and it is this value that is passed out of the macro. ⑥ When the directory already exists, the full path is returned at this point in the macro.

⑦ Because %CHECKLOC is itself written as a function that returns an existing path, it can be used in a LIBNAME statement. Here %CHECKLOC establishes the path `c:\temp\test` (if it does not already exist), and it creates the `libref TEMTEST`. 

```libname temtest "%checkloc(dirloc=c:\temp, dirname=test)"; ⑦```
When you want to access the names of the files in a directory you will need to open and close the directory using the DOPEN and DCLOSE functions. Once the directory is opened, you can use the DREAD and DNUM functions to step through the files in the directory. The %FILLIST macro writes the names of all the SAS programs in a directory to the SAS Log.

Establishing a process for each file in a directory would be a fairly straightforward expansion of the %FILLIST macro. Since the %DO loop will execute once for each file in the directory, any process that depends on the name of the file will have sufficient information.

In the next macro %FILLIST is expanded in a couple of important ways. The critical elements of %FILLIST remain, however in this macro, instead of just a %PUT, we include a process (to convert all CSV files in a directory to SAS data sets using PROC IMPORT). Not only do we need to convert the CSV files in the current directory, but in all subdirectories as well. This can be accomplished using a technique known as recursion.

A recursive macro is a macro that calls itself. The macro %FILLIST can be expanded to search for all files of a given type across a directory, including sub-directories, by making it recursive. The macro %RECURSIVEIMPORTDATA shown here is a simplified version of a macro written by Phuong Dinh of Cornerstone Research Inc. A similar, albeit even more simplistic, version of this macro appears in the SAS 9.4 Macro Language reference manual. Ostensibly this macro converts all CSV files in a folder (and subfolders) to SAS data sets and appends them into a single data set, however the important take away is that it uses recursion and a series of directory and file functions to search the sub-directories as well.

In this macro each file in a directory is examined by passing the macro the path of the directory of interest. If the file is a CSV file it is converted to a SAS data set, however if the file is a subdirectory, the macro is called again, this time with the path to the subdirectory. Because it is recursively called, this macro will crawl through an entire directory including all levels of subdirectories.

```sas
%macro filList(filer=);
%local rc fid i fname;
%let fid = %sysfunc(dopen(&filer)); ③
%if &fid %then %do i = 1 %to %sysfunc(dnum(&fid)); ⑨
   %let fname= %sysfunc(dread(&fid,&i)); ⑩
   %if %upcase(%qscan(&fname,-1,.))=SAS %then %put &fname;
%end;
%let fid = %sysfunc(dclose(&fid)); ⑪
%mend fillist;

filename saspgms 'c:\temp';
%fillist(filerf=saspgms)
filename saspgms clear;
```

③ The directory specified in the fileref passed into the macro is opened for use. The directory identification number is saved in the macro variable &FID.
⑨ The DNUM function returns the number of files in the directory.
⑩ The name of the &Ith file is returned by the DREAD function, and the names are written to the SAS Log.
⑪ After using the directory, it is closed.
A unique fileref name is created. A level counter is used so that when the macro is called recursively the fileref created by the inner macro will not replace a fileref that already exists. Because a fileref is restricted to 8 characters, this macro cannot accommodate more than 1,000 levels (the highest level defaults to &LEVELCOUNTER=0).

The FILEEXIST function is used to check to make sure that the requested folder exists. If it does not exist a custom error message is written to the SAS Log and the macro terminates execution.

The FILENAME function is used to establish the fileref for this folder. Notice that although the name of the fileref is stored in &_REF, the ampersand is not used with the FILENAME function.

Assuming that the fileref (&_REF) is successfully established (&_RC=0), the folder designated by the fileref &_REF is opened for processing. The folder’s identification number is saved in &_DSID.
When the directory is successfully opened (%_DSID ne 0), a %DO loop is used to step through all of the files in the current folder. The number of files to be processed is returned by the DNUM function. The index of the %DO loop (&_I) will be used by the DREAD function.

The DREAD function is used to read the name of the &_Ith file. The file’s name is stored in the macro variable &_FILENAME.

The extension of the current file name is extracted using the %SCAN function. If there is no extension the name of the file will be returned. This macro assumes that all files except folders have extensions.

The file name has an extension of CSV. Import the CSV file, create a data set and append it to the growing data table. This file does not have an extension and is therefore assumed to be a sub-directory. The macro %RECURSIVEIMPORTDATA is called with the name of the subfolder and with a level indicator increased by 1.

The directory is closed using the DCLOSE function.

The fileref for this folder is cleared. Remember that the macro variable &_REF is specified without the ampersand in the Filename function.

RELATED FUNCTIONS AND OTHER TOOLS
There are a number of less commonly used directory functions. Remember less commonly used does not necessarily mean less useful. These are most useful when you want to handle the directory as an entity.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Action:</th>
</tr>
</thead>
<tbody>
<tr>
<td>DINFO</td>
<td>Returns information about a directory</td>
</tr>
<tr>
<td>DOPTNAME</td>
<td>Returns directory attribute information</td>
</tr>
<tr>
<td>DOPTNUM</td>
<td>Returns the number of attribute items</td>
</tr>
</tbody>
</table>

Table 6: Other Directory Functions

The macro %DIRINFO can be used to show the directory information returned by the functions in Table 6. It has been my experience that, under Windows at least, only a limited amount of information is returned. It is possible that other operating systems and directory structures will return more useful information. Experiment - execute %DIRINFO on your OS.
Much like the functions that can be used to read and write data in a SAS data set, there are also a number of similar functions that can be used to read and write information to and from non-SAS controlled files. Some of those functions are shown in Table 7.

<table>
<thead>
<tr>
<th>Function Name</th>
<th>Action:</th>
</tr>
</thead>
<tbody>
<tr>
<td>FOPEN and FCLOSE</td>
<td>Opens and closes a specific file</td>
</tr>
<tr>
<td>FREAD</td>
<td>Reads a row into the File Data Buffer (FDB)</td>
</tr>
<tr>
<td>FAPPEND</td>
<td>Appends the current record to an existing file</td>
</tr>
<tr>
<td>FCOL</td>
<td>Current position on the FDB</td>
</tr>
<tr>
<td>FGET</td>
<td>Retrieves item from the FDB</td>
</tr>
<tr>
<td>FWRITE</td>
<td>Writes a record to an external file</td>
</tr>
<tr>
<td>FDELETE</td>
<td>Deletes an external file</td>
</tr>
<tr>
<td>FPOINT</td>
<td>Contains number of the next row to read</td>
</tr>
</tbody>
</table>

Table 7: File I/O Functions

SUMMARY
There are a number of SAS DATA step functions that you will likely never use in the DATA step. However when paired with the macro language through the use of the %SYSFUNC macro function, these DATA step functions have extensive utility. Functions that are designed to work with the metadata of SAS data sets and OS folders are especially valuable. These functions allow us to retrieve, manipulate, and use information that in many cases would be otherwise unavailable to us; and we do not need to leave the macro language in order to take advantage of them!

ABOUT THE AUTHOR
Art Carpenter is a SAS Certified Advanced Professional Programmer and his publications list includes; five books and numerous papers and posters presented at SAS Global Forum, SUGI, PharmaSUG, WUSS, and other regional conferences. Art has been using SAS since 1977 and has served in various leadership positions in local, regional, and international user groups.

Recent publications are listed on my sasCommunity.org Presentation Index page. 
http://sascommunity.org/wiki/Presentations:ArtCarpenter_Papers_and_Presentations

AUTHOR CONTACT
Arthur L. Carpenter
California Occidental Consultants
10606 Ketch Circle
Anchorage, AK 99515

(907) 865-9167
art@caloxy.com
www.caloxy.com
REFERENCES


There are a number of nice examples in the SAS 9.4 Macro Language Reference manual that have to do with reading files within directories, start here: http://support.sas.com/documentation/cdl/en/mcrolref/69726/HTML/default/viewer.htm#n02xowj8yuqfo4n0zzi98shu8qup.htm

TRADEMARK INFORMATION
SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries.

® indicates USA registration.

Other brand and product names are trademarks of their respective companies.